VBA Macro for Excel 2003 Export of Text File with Comma and Quote Delimiters.

This code is a sample from the MSDN script library at http://msdn.microsoft.com/library/default.asp?url=/library/en-us/odc_xl2003_ta/html/odc_XL_Samples.asp
Or http://msdn.microsoft.com/en-us/library/Aa203726.aspx
1. Open a new workbook. 

2. On the Tools menu, point to Macro, and then click Visual Basic Editor (or simply press ALT+F11). In the Visual Basic Editor, click the Insert menu and then Module. 

3. Type or paste the below sample code into the module.

4. Before running the macro, select the data that you want to export, and then point to Macros on the Tools menu and click Macro. 

5. Select the QuoteCommaExport macro, and click Run. 

Sample code:
Sub QuoteCommaExport()

    Dim DestFile As String

    Dim FileNum As Integer

    Dim ColumnCount As Integer

    Dim RowCount As Integer

    ' Prompt user for destination file name.

    DestFile = InputBox("Enter the destination filename" & _

      Chr(10) & "(with complete path and extension):", _

      "Quote-Comma Exporter")

    ' Obtain next free file handle number.

    FileNum = FreeFile()

    ' Turn error checking off.

    On Error Resume Next

    ' Attempt to open destination file for output.

    Open DestFile For Output As #FileNum

    ' If an error occurs report it and end.

    If Err <> 0 Then

      MsgBox "Cannot open filename " & DestFile

      End

    End If

    ' Turn error checking on.

    On Error GoTo 0

    ' Loop for each row in selection.

    For RowCount = 1 To Selection.Rows.Count

      ' Loop for each column in selection.

      For ColumnCount = 1 To Selection.Columns.Count

         ' Write current cell's text to file with quotation marks.

         Print #FileNum, """" & Selection.Cells(RowCount, _

            ColumnCount).Text & """";

         ' Check if cell is in last column.

         If ColumnCount = Selection.Columns.Count Then

            ' If so, then write a blank line.

            Print #FileNum,

         Else

            ' Otherwise, write a comma.

            Print #FileNum, ",";

         End If

      ' Start next iteration of ColumnCount loop.

      Next ColumnCount

    ' Start next iteration of RowCount loop.

    Next RowCount

    ' Close destination file.

    Close #FileNum

End Sub

